



SPARQL - Querying the Web of Data

Seminar WS 2008/2009

An Introduction to SPARQL


Olaf Hartig
hartig@informatik.hu-berlin.de

SPARQL in General

- SPARQL Protocol and RDF Query Language
- SPARQL Query Language for RDF
 - Declarative
 - Based on the RDF data model (triples/graph)
 - Our focus
- SPARQL Query Results XML Format
 - Representation of the results of SPARQL queries
- SPARQL Protocol for RDF
 - Transmission of SPARQL queries and the results
 - SPARQL endpoint: Web service that implements the protocol


Main Idea of SPARQL Queries

- Main idea: pattern matching
 - Describe subgraphs of the queried RDF graph
 - Subgraphs that match your description yield a result
 - Mean: graph patterns (i.e. RDF graphs with variables)



Main Idea of SPARQL Queries

Queried
graph:



Results:

?v
dbpedia:Mount_Baker
dbpedia:Mount_Etna



Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```



Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- Prologue:
 - **Prefix definitions enable CURIEs in the query**
 - **Attention: No period (“.”) character to separate (as in N3)**

Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Result form specification:**
 - **SELECT, DESCRIBE, CONSTRUCT, or ASK**
 - **SELECT:** - Variable list or asterisk (“*”) character for all
 - **DISTINCT** for disjoint results



Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Dataset specification:**
 - Specify the dataset to be queried
 - Use **FROM** and **FROM NAMED** clauses (each with a URI)



Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Query Pattern:**
 - **WHERE clause specifies the graph pattern to be matched**



Components of SPARQL Queries

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Solution modifiers:**
 - Modify the result set, but not the single results
 - ORDER BY, LIMIT, or OFFSET

Graph Patterns

- Different **types of graph patterns for the query pattern (WHERE clause):**
 - Basic graph pattern (BGP)
 - Group graph pattern
 - Optional graph pattern
 - Union graph pattern
 - Graph graph pattern
 - (Constraints)

Basic Graph Patterns



- Set of triple patterns (i.e. RDF triples with variables)
- Variable names prefixed with “?” or “\$” (e.g. ?v, \$v)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?name
WHERE {
    ?v rdf:type umbel-sc:Volcano .
    ?v rdfs:label ?name .
}
```

Basic Graph Patterns

- Set of triple patterns (i.e. RDF triples with variables)
- Variable names prefixed with “?” or “\$” (e.g. ?v, \$v)
- Turtle syntax (similar to N3)
- Syntactical sugar as in N3 (e.g. property and object lists)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?name
WHERE {
    ?v rdf:type umbel-sc:Volcano ;
        rdfs:label ?name .
}
```

Basic Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                     umbel-sc:NaturalElevation ;  
                     rdfs:label "Beerenberg"@en ;  
                     rdfs:label "Бееренберг"@ru .
```

- Question: What are the names of all (known) volcanos?

```
SELECT ?name WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
       rdfs:label ?name . }
```

Query*

Result:

?name
"Etna"
"Бееренберг"@ru
"Beerenberg"@en

*Prefix definitions omitted

Basic Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .

dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.

dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

- Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {
  ?v rdf:type ?type ;
      rdfs:label "Beerenberg" .
}
```

Query

?type

Basic Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

- Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {
  ?v rdf:type ?type ;
  rdfs:label "Beerenberg"@en .
}
```

Query

?type
umbel-sc:Volcano
umbel-sc:NaturalElevation

Basic Graph Patterns

```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ; Data  
          p:location dbpedia:United_States .  
dbpedia:United_States rdfs:label "United States" .
```

- Question: Where are all (known) volcanos located? List the names.
- Blank nodes in SPARQL queries
 - Permitted as subject and object of a triple pattern
 - Like non-selectable variables

```
SELECT ?name WHERE {  
  _:x rdf:type umbel-sc:Volcano ;  
        p:location [ rdfs:label ?name ] . }
```

Query

?name
"United States"

Basic Graph Patterns

```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ; Data
  p:location [ rdfs:label "United States"@en ,
    "États-Unis"@fr ] .

dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;
  p:location [ rdfs:label "Italy" ] .
```

- Blank nodes in the queried graph
 - Blank node identifiers may occur in the results

```
SELECT ?l ?name WHERE {
  ?v rdf:type umbel-sc:Volcano ;
    p:location ?l .
  ?l rdfs:label ?name .
}
```

Query

?l	?name
_:x	"United States"@en
_:x	"États-Unis"@fr
_:y	"Italy"

Optional Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Etna" .
```

Data

```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Beerenberg"@en .
```

- Question: What are all (known) volcanos and their names?

```
SELECT ?v ?name WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
       rdfs:label ?name . }
```

Query

- Problem: Mount Baker missing (it has no name)

?v	?name
dbpedia:Mount_Etna	"Etna"
dbpedia:Beerenberg	"Beerenberg"@en

Optional Graph Patterns

- Keyword **OPTIONAL** allows for optional patterns

```
SELECT ?v ?name WHERE {  
    ?v rdf:type umbel-sc:Volcano .  
    OPTIONAL { ?v rdfs:label ?name }  
}
```

Query

?v	?name
dbpedia:Mount_Etna	"Etna"
dbpedia:Mount_Baker	
dbpedia:Beerenberg	"Beerenberg"@en

- Optional patterns may yield unbound variables

Union Graph Patterns

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Etna" ;  
                     p:location dbpedia:Italy .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
                     p:location dbpedia:United_States .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Beerenberg"@en ;  
                     p:location dbpedia:Norway .
```

- **Question: What volcanos are located in the Italy or in Norway?**

```
SELECT ?v WHERE {  
    ?v rdf:type umbel-sc:Volcano ;  
         p:location ? . }
```

Query

Union Graph Patterns



- Union graph patterns allow for alternatives

Semantically equivalent

```
SELECT ?v WHERE {  
  { ?v rdf:type umbel-sc:Volcano ;  
    p:location dbpedia:Italy }  
UNION  
  { ?v rdf:type umbel-sc:Volcano ;  
    p:location dbpedia:Norway }  
}
```

Query

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
  { ?v p:location dbpedia:Italy }  
UNION  
  { ?v p:location dbpedia:Norway }  
}
```

Query

Group Graph Patterns

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
                  { ?v p:location dbpedia:Italy }  
UNION  
                  { ?v p:location dbpedia:Norway }  
}
```

Query

Semantically equivalent



```
SELECT ?v WHERE { { ?v rdf:type umbel-sc:Volcano }  
                  { { ?v p:location dbpedia:Italy }  
UNION  
                  { ?v p:location dbpedia:Norway } }  
}
```

Query

Constraints

- Constraints filter solutions
- Keyword **FILTER** followed by expression
- Filter expressions contain operators and functions

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
PREFIX p: <http://dbpedia.org/property/>

SELECT ?v
WHERE {
    ?v rdf:type umbel-sc:Volcano ;
        p:lastEruption ?le .
    FILTER ( ?le > 1900 )
}
```

Constraints

- Operators and functions operate on RDF terms
- Filter expressions evaluate to true, false, or error
- Truth table:

A	B	A B	A && B
T	T	T	T
T	F	T	F
F	T	T	F
F	F	F	F
T	E	T	E
E	T	T	E
F	E	E	F
E	F	E	F
E	E	E	E

Constraints

- Unary operators:

Operator	Type(A)	Result type
! A	xsd:boolean	xsd:boolean
+ A	numeric	numeric
- A	numeric	numeric
BOUND(A)	variable	xsd:boolean
isURI(A)	RDF term	xsd:boolean
isBLANK(A)	RDF term	xsd:boolean
isLITERAL(A)	RDF term	xsd:boolean
STR(A)	literal / URI	simple literal
LANG(A)	literal	simple literal
DATATYPE(A)	literal	simple literal

Constraints

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

- Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {
  ?v rdf:type ?type ;
      rdfs:label ?name .
  FILTER ( STR(?name) = "Beerenberg" )
}
```

	?type
	umbel-sc:Volcano
	umbel-sc:NaturalElevation

Constraints

- **Binary operators:**
 - Logical connectives `&&` and `||` for `xsd:boolean`
 - Comparison operators `=`, `!=`, `<`, `>`, `<=`, and `>=` for numeric datatypes, `xsd:dateTime`, `xsd:string`, and `xsd:boolean`
 - Comparison operators `=` and `!=` for other datatypes
 - Arithmetic operators `+`, `-`, `*`, and `/` for numeric datatypes
- **Furthermore:**
 - `REGEX(String,Pattern)` or `REGEX(String,Pattern,Flags)`
 - `sameTERM(A,B)`
 - `langMATCHES(A,B)`

Constraints

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

- Question: What volcanos have an “e” in their name?

```
SELECT ?v WHERE {
  ?v rdf:type umbel-sc:Volcano ;
      rdfs:label ?name .
  FILTER( REGEX(STR(?name), "e") )
}
```

Query

?v
dbpedia:Beerenberg
dbpedia:Beerenberg

Constraints

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

- Question: What volcanos have an “e” in their name?

```
SELECT ?v WHERE {
  ?v rdf:type umbel-sc:Volcano ;
      rdfs:label ?name .
  FILTER( REGEX(STR(?name),"e","i") )
}
```

Query

?v
dbpedia:Mount_Etna
dbpedia:Beerenberg
dbpedia:Beerenberg

Negation

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Beerenberg"@en ;  
                     rdfs:label "Бееренберг"@ru .
```

Data

- Question: What volcanos do **not** have a name in our data?

```
SELECT ?v WHERE {
```

Query

```
?v rdf:type umbel-sc:Volcano .  
OPTIONAL { ?v rdfs:label ?name }  
FILTER( ! BOUND(?name) )
```

```
}
```

?v

dbpedia:Mount_Baker

Negation

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Beerenberg"@en ;  
                     rdfs:label "Бееренберг"@ru .
```

Data

- Question: What volcanos are **not** called “Beerenberg”?

~~SELECT ?v WHERE {
 ?v rdf:type umbel-sc:Volcano .
 ?v rdfs:label ?name .
 FILTER (STR(?name) != "Beerenberg")
}~~

Query

\neq

?v
dbpedia:Mount_Etna
dbpedia:Mount_Baker
dbpedia:Beerenberg

Negation

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Etna" .
```

Data

```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .
```

```
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                     rdfs:label "Beerenberg"@en ;  
                     rdfs:label "Бееренберг"@ru .
```

- Question: What volcanos are **not** called “Beerenberg”?

```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  OPTIONAL { ?v rdfs:label ?name .  
             FILTER (STR(?name) = "Beerenberg") }  
  FILTER ( ! BOUND(?name) )  
}
```

?v
dbpedia:Mount_Etna
dbpedia:Mount_Baker

Negation as Failure

Graph Graph Patterns

- SPARQL queries are executed against an RDF dataset
- An **RDF dataset** comprises:
 - One **default graph** and
 - Zero or more **named graphs** (identified by an URI)
- Keyword **GRAPH** makes one of the named graphs the **active graph** used for pattern matching

Graph Graph Patterns

```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.
```

Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1  
    rdf:type umbel-sc:Volcano ;  
    rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
    rdf:type umbel-sc:Volcano .
```

```
dbpedia:Beerenberg http://example.org/d3  
    rdf:type umbel-sc:Volcano ;  
    rdfs:label "Beerenberg"@en .
```

Graph Graph Patterns

```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.
```

```
dbpedia:Mount_Etna http://example.org/d1  
    rdf:type umbel-sc:Volcano ;  
    rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
    rdf:type umbel-sc:Volcano .
```

```
SELECT ?v WHERE {  
  GRAPH <http://example.org/d1> {  
    ?v rdf:type umbel-sc:Volcano .  
  }  
}  
?v  
dbpedia:Mount_Etna
```

Graph Graph Patterns

```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.
```

```
dbpedia:Mount_Etna http://example.org/d1  
    rdf:type umbel-sc:Volcano ;  
    rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
    rdf:type umbel-sc:Volcano .
```

```
SELECT ?v WHERE {
```

```
GRAPH ?g {
```

```
    ?v rdf:type umbel-sc:Volcano .
```

```
}
```

```
}
```

```
?v
```

```
dbpedia:Mount_Etna  
dbpedia:Mount_Baker  
dbpedia:Beerenberg
```

Graph Graph Patterns

`dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.`
`dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.`

`dbpedia:Mount_Etna http://example.org/d1`
 `rdf:type umbel-sc:Volcano ;`
 `rdfs:label "Etna" .`

`dbpedia:Mount_Baker http://example.org/d2`
 `rdf:type umbel-sc:Volcano .`

`SELECT ?v ?g WHERE {`

`GRAPH ?g {`
 `?v rdf:type`
 `}`
}

<code>?v</code>	<code>?g</code>
<code>dbpedia:Mount_Etna</code>	<code><http://example.org/d1></code>
<code>dbpedia:Mount_Baker</code>	<code><http://example.org/d2></code>
<code>dbpedia:Beerenberg</code>	<code><http://example.org/d3></code>

Graph Graph Patterns

```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.
```

```
dbpedia:Mount_Etna http://example.org/d1  
    rdf:type umbel-sc:Volcano ;  
    rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
    rdf:type umbel-sc:Volcano .
```

```
SELECT ?v WHERE {  
  _:x rdfs:seeAlso ?g  
  GRAPH ?g {  
    ?v rdf:type umbel-sc:Volcano .  
  }  
}
```

?v

```
dbpedia:Mount_Etna  
dbpedia:Mount_Baker
```

Graph Graph Patterns

`dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.`

`dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.`

`dbpedia:Mount_Etna http://example.org/d1`

`rdf:type umbel-sc:Volcano ;`

`rdfs:label "Etna" .`

`dbpedia:Mount_Baker http://example.org/d2`

`rdf:type umbel-sc:Volcano .`

`dbpedia:Beerenberg http://example.org/d3`

`rdf:type umbel-sc:Volcano ;`

`rdfs:label "Beerenberg"@en .`

- Question: Which named graphs contain the name of a volcano that is not referenced in the default graph?

Graph Graph Patterns

```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1>.  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2>.
```

```
dbpedia:Mount_Etna http://example.org/d1  
    rdf:type umbel-sc:Volcano ;  
    rdfs:label "Etna" .
```

```
SELECT ?g WHERE {  
  GRAPH ?g {  
    ?v rdf:type umbel-sc:Volcano ;  
    rdfs:label ?name .  
  }  
  OPTIONAL { ?v rdfs:seeAlso ?r }  
  FILTER ( ! BOUND(?r) )  
}
```

```
http://example.org/d2  
sc:Volcano .
```

```
http://example.org/d3  
-sc:Volcano ;  
erenberg"@en .
```

ain the name of a
he default graph?

Summary – Graph Patterns

- Different **types of graph patterns** for the query pattern (**WHERE clause**):
 - Basic graph pattern (**BGP**)
 - Group graph pattern
 - Optional graph pattern – keyword **OPTIONAL**
 - Union graph pattern – keyword **UNION**
 - Graph graph pattern – keyword **GRAPH**
 - Constraints – keyword **FILTER**

Result Forms



- **SELECT**
 - Sequence of results (i.e. sets of variable bindings)
 - Selected variables separated by space (not by comma!)
 - Asterisk (“*”) character selects all variables in the pattern
- **ASK**
 - Check if there is at least one result
 - Example: Do we have volcanos that do not have a name?

```
ASK WHERE { ?v rdf:type umbel-sc:Volcano .  
           OPTIONAL { ?v rdfs:label ?name }  
           FILTER( ! BOUND(?name) )  
 }
```

Query

Result Forms

- **DESCRIBE**
 - Returns an RDF graph with data about resources
 - Nondeterministic (i.e. query processor determines the actual structure of the returned RDF graph)
 - Name the resource:

```
DESCRIBE <http://dbpedia.org/resource/Beerenberg> Query
```

- Specify the resource with a query pattern:

```
DESCRIBE ?v WHERE {  
    ?v rdf:type umbel-sc:Volcano ; rdfs:label ?name .  
    FILTER ( STR(?name) = "Beerenberg" ) }
```

- Multiple variables possible or asterisk ("*") for all

Result Forms



- **CONSTRUCT**
 - Returns an RDF graph created from a template
 - Template: graph pattern with variables from the query pattern

```
CONSTRUCT { ?v rdfs:label ?name ;                                     Query
            rdf:type myTypes:VolcanosOutsideTheUS }

WHERE {
    ?v rdf:type umbel-sc:Volcano ;
        rdfs:label ?name .

OPTIONAL { ?v p:location ?l
            FILTER ( ?l = dbpedia:United_States ) }

FILTER ( ! BOUND(?l) )

}
```

Result Forms

dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
rdfs:label "Etna" ;
p:location dbpedia:Italy .

dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;
rdfs:label "Mount Baker" ;
p:location dbpedia:United_States .

dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;
rdfs:label "Beerenberg"@en ;
p:location dbpedia:Norway .

dbpedia:Mount_Etna rdfs:label "Etna" ; Result
rdf:type myTypes:VolcanosOutsideTheUS .

dbpedia:Beerenberg rdf:type myTypes:VolcanosOutsideTheUS ;
rdfs:label "Beerenberg"@en .

Solution Modifiers

- **Modify the result set, but not the single results**
- **Permitted for SELECT queries only**
 - **DISTINCT**
 - **ORDER BY**
 - **LIMIT**
 - **OFFSET**

Solution Modifiers

- **DISTINCT** – removes duplicates from the result set

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

```
SELECT ?type
```

```
WHERE { _:x rdf:type ?type }
```

Query

?type

umbel-sc:Volcano
umbel-sc:Volcano
umbel-sc:NaturalElevation
umbel-sc:Volcano

Solution Modifiers

- **DISTINCT** – removes duplicates from the result set

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ; Data
                     rdfs:label "Etna" .
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,
                     umbel-sc:NaturalElevation ;
                     rdfs:label "Beerenberg"@en ;
                     rdfs:label "Бееренберг"@ru .
```

```
SELECT DISTINCT ?type          Query
WHERE { _:x rdf:type ?type }
```

?type
umbel-sc:Volcano
umbel-sc:NaturalElevation

Solution Modifiers

- **ORDER BY** – orders the results

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano ; Query  
                  rdfs:label ?name }  
  
ORDER BY ?name
```

- Order for different kinds of elements:
 unbound variable < blank node < URI < literal
- **ASC** for ascending (default) and **DESC** for descending
- Hierarchical order criteria:

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ; Query  
                  p:lastEruption ?le ;  
                  rdfs:label ?name }  
  
ORDER BY DESC(?le), ?name
```

Solution Modifiers

- **LIMIT** – limits the number of results

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ; Query
                      rdfs:label ?name }

ORDER BY ?name

LIMIT 5
```

- **OFFSET** – position/index of the first returned results

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ; Query
                      rdfs:label ?name }

ORDER BY ?name

LIMIT 5 OFFSET 10
```

- Only useful if the order is predictable (i.e. ordered results)

Further Reading

- W3C RDF Data Access Working Group
<http://www.w3.org/2001/sw/DataAccess/>
 - SPARQL Query Language for RDF
 - SPARQL Protocol for RDF
 - SPARQL Query Results XML Format
- SPARQL interface
for dbpedia:
<http://dbpedia.org/snorql/>

SPARQL:

```
PREFIX owl: <http://www.w3.org/2002/07/owl#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX foaf: <http://xmlns.com/foaf/0.1/>
PREFIX dct: <http://purl.org/dc/elements/1.1/>
PREFIX l: <http://dbpedia.org/resource/>
PREFIX dbpedia2: <http://dbpedia.org/property/>
PREFIX dbpedia: <http://dbpedia.org/>
PREFIX skos: <http://www.w3.org/2004/02/skos/core#>

SELECT * WHERE {
    ?p rdf:type ?t1; rdf:type ?t2; rdfs:label "Beerenberg" .
    FILTER (?t1 != ?t2)
}
```

Results: [Browse](#) [Go!](#) [Reset](#)

SPARQL results:

p	t	t2
:Beerenberg	dbpedia:class/yago/RidgeVolcanoes	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	dbpedia:class/yago/VolcanoesOfNorway	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	<http://umbel.org/umbel/sc/NaturalElevation>	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	<http://umbel.org/umbel/sc/LandTopographicalFeature>	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	<http://umbel.org/umbel/sc/Mountain>	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	<http://umbel.org/umbel/sc/Volcano>	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	dbpedia:ontology/Resource#Class	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	dbpedia:ontology/Place#Class	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	dbpedia:ontology/Mountain#Class	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean
:Beerenberg	dbpedia:class/yago/VolcanoesOfTheAtlanticOcean	dbpedia:class/yago/RidgeVolcanoes
:Beerenberg	dbpedia:class/yago/VolcanoesOfNorway	dbpedia:class/yago/RidgeVolcanoes
-Rosenbaum	<http://dbpedia.org/resource/NaturalFormation>	dbpedia:resource/face/summitOfRidgeVolcanoes